```
animate = $t * 10;
steps = 200;
r_cup = 32.5;
r_max = 33.0;
// Gerar valores aleatorios para os parametros
r_rand = rands(r_cup, r_max, steps);
h_rand = rands(0.2, 1.0, steps); // Garantir que todas as alturas sao positivas
rot_rand = rands(0, 360, steps);
fn_rand = rands(9, 12, steps);

// Criar array acumulado de alturas com limite de 90 mm
function cumulative_heights(arr, sum = 0, index = 0) =
    (sum >= 90 || index >= len(arr)) ? [90] :
    let(next_sum = sum + arr[index])
    concat([sum], cumulative_heights(arr, (next_sum > 90) ? 90 : next_sum, index + 1));

h_total = cumulative_heights(h_rand);
// Construcao do modelo
color("grey") {
    for (i = [0:len(h_total)-4]) {
        let(
            z1 = h_total[i],       z2 = h_total[i+1],
            z3 = h_total[i+2],     z4 = h_total[i+3],
            r1 = r_rand[i],        r2 = r_rand[i+1],
            r3 = r_rand[i+2],      r4 = r_rand[i+3],
            h1 = h_rand[i],        h2 = h_rand[i+1],
            h3 = h_rand[i+2],      h4 = h_rand[i+3],
            fn1 = fn_rand[i],      fn2 = fn_rand[i+1],
            fn3 = fn_rand[i+2],    fn4 = fn_rand[i+3],
            rot1 = rot_rand[i],    rot2 = rot_rand[i+1],
            rot3 = rot_rand[i+2],  rot4 = rot_rand[i+3])
        hull() {
            if (i == 0) {
                translate([0, 0, 0])
                    cylinder(r = r1, h = 1, $fn = 180);
            }

            rotate([0, 0, rot1])
                translate([0, 0, z1])
                    cylinder(h = h1, r = r1, $fn = fn1);

            rotate([0, 0, rot2])
                translate([0, 0, z2])
                    cylinder(h = h2, r = r2, $fn = fn2);

            rotate([0, 0, rot3])
                translate([0, 0, z3])
                    cylinder(h = h3, r = r3, $fn = fn3);

            rotate([0, 0, rot4])
                translate([0, 0, z4])
                    cylinder(h = h4, r = r4, $fn = fn4);
            if (i == len(h_total) - 4) {
                translate([0, 0, 90])
                    cylinder(r = r4, h = 1.5, $fn = 180);
            }
        }
    }
}
```